

SAS Tips and Tricks

Leonard Landry
Statistics Canada

SAS Tips and Tricks

- End of File processing when subsetting a dataset.
 - Read an entire file and perform some processing after the last record is read.
 - Problems may result from improper placement of the end of file processing code

End of File Processing and Subsetting Records

Obs	Sex	VarA
1	M	10
2	F	5
3	M	20
4	F	15
5	M	15
6	F	5
7	M	12
8	F	20

End of File Processing and Subsetting Records

```
Data _null_;
```

```
  set test end=end1;
```

```
  retain total 0;
```

```
  total = total + VarA;
```

```
  if end1 then put total=;
```

```
run;
```

End of File Processing and Subsetting Records

total=102

NOTE: There were 8 observations read
from the data set WORK.TEST.

NOTE: DATA statement used (Total
process time):

real time	0.00 seconds
cpu time	0.00 seconds

End of File Processing and Subsetting Records

```
Data _null_;  
  set test end=end1;  
  if sex = 'F';  
  retain total 0;  
  total = total + varA;  
  if end1 then put total=;  
run;
```

End of File Processing and Subsetting Records

total=45

NOTE: There were 8 observations read
from the data set WORK.TEST.

NOTE: DATA statement used (Total
process time):

real time	0.00 seconds
cpu time	0.00 seconds

End of File Processing and Subsetting Records

```
Data _null_;  
  set test end=end1;  
  if sex = 'M';  
  retain total 0;  
  total = total + varA;  
  if end1 then put total=;  
run;
```



End of File Processing and Subsetting Records

NOTE: There were 8 observations read
from the data set WORK.TEST.

NOTE: DATA statement used (Total
process time):

real time	0.00 seconds
cpu time	0.00 seconds

End of File Processing and Subsetting Records

Obs	Sex	VarA
1	M	10
2	F	5
3	M	20
4	F	15
5	M	15
6	F	5
7	M	12
8	F	20

End of File Processing and Subsetting Records

- The problem occurs when the last record is deleted by the **SUBSETTING IF**.
- When a record is deleted by a subsetting if SAS stops processing and returns to the next iteration of the data step.

End of File Processing and Subsetting Records

```
Data _null_;  
  set test end=end1;  
  where sex = 'M';  
  retain total 0;  
  total = total + varA;  
  if end1 then put total=;  
run;
```

End of File Processing and Subsetting Records

total=57

NOTE: There were 4 observations read
from the data set WORK.TEST.

NOTE: DATA statement used (Total
process time):

real time	0.00 seconds
cpu time	0.00 seconds

End of File Processing and Subsetting Records

```
Data _null_;
```

```
if end1 then put total=;
```

```
set test end=end1;
```

```
if sex = 'M';
```

```
retain total 0;
```

```
total = total + varA;
```

```
run;
```

End of File Processing and Subsetting Records

total=57

NOTE: There were 8 observations read
from the data set WORK.TEST.

NOTE: DATA statement used (Total
process time):

real time	0.00 seconds
cpu time	0.00 seconds

End of File Processing and Subsetting Records

```
Data _null_;  
if end1 then put total=;  
Infile 'c:\sastests\test' recfm=f lrecl=4 end=end1;  
input @1 sex $1.  
      @3 VarA 2.;  
if sex = 'M';  
retain total 0;  
total = total + varA;  
run;
```

End of File Processing and Subsetting Records

total=57

NOTE: 8 records were read from the infile
'c:\sastests\test1'.

NOTE: DATA statement used (Total
process time):

real time	0.00 seconds
cpu time	0.00 seconds

End of File Processing and Subsetting Records

FileA		FileB	
ID	VarA	ID	VarB
1	10	1	5
2	15	2	8
3	12	4	25
5	20	5	11

End of File Processing and Subsetting Records

```
Data FileAB(keep=ID varA varB)
sums(keep=SumvarA SumvarB);
retain SumvarA SumvarB 0;
merge FileA(in=inA) FileB(in=inB) end=end1;
by ID;
if inA & inB;
SumvarA = SumvarA + VarA;
SumvarB = SumvarB + VarB;
output FileAB;
if end1 then output sums;
Run;
```

End of File Processing and Subsetting Records

NOTE: There were 4 observations read from the data set WORK.FILEA.

NOTE: There were 4 observations read from the data set WORK.FILEB.

NOTE: The data set WORK.FILEAB has 3 observations and 3 variables.

NOTE: The data set WORK.SUMS has 1 observations and 2 variables.

NOTE: DATA statement used (Total process time):

real time 0.01 seconds

cpu time 0.01 seconds

End of File Processing and Subsetting Records

FileA		FileB	
ID	VarA	ID	VarB
1	10	1	5
2	15	2	8
3	12	4	25
5	20	5	11
		6	20

End of File Processing and Subsetting Records

NOTE: There were 4 observations read from the data set WORK.FILEA.

NOTE: There were 5 observations read from the data set WORK.FILEB.

NOTE: The data set WORK.FILEAB has 3 observations and 3 variables.

NOTE: The data set WORK.SUMS has 0 observations and 2 variables.

NOTE: DATA statement used (Total process time):

real time	0.01 seconds
cpu time	0.01 seconds



End of File Processing and Subsetting Records

```
Data FileAB(keep=ID varA varB)
  sums(keep=SumvarA SumvarB);
  retain SumvarA SumvarB 0;
  if end1 then output sums;
  merge FileA(in=inA) FileB(in=inB) end=end1;
  by ID;
  if inA & inB;
  SumvarA = SumvarA + VarA;
  SumvarB = SumvarB + VarB;
  output FileAB;

Run;
```

End of File Processing and Subsetting Records

NOTE: There were 4 observations read from the data set WORK.FILEA.

NOTE: There were 5 observations read from the data set WORK.FILEB.

NOTE: The data set WORK.FILEAB has 3 observations and 3 variables.

NOTE: The data set WORK.SUMS has 1 observations and 2 variables.

NOTE: DATA statement used (Total process time):

real time 0.01 seconds

cpu time 0.01 seconds



Summary

- Improper placement of the end of file processing code may result in the code not being executed.
- The problem does not occur when using a WHERE Statement.
- Placing the end of file processing code before the set statement works.
- With the MERGE statement placing the end of file processing code before the merge statement works.



Questions / Comments



Statistics
Canada

Statistique
Canada

Greg McLean

Project Leader - SAS Technology Centre
System Development Division
R.H. Coats Building, 14th Floor, Section Q

Ottawa, Ontario, Canada K1A 0T6
(613) 951-2396 Fax (613) 951-0607

greg.mclean@statcan.ca

Canada



Statistics
Canada

Statistique
Canada

Leonard Landry

Database Manager
Business and Labour Market Analysis Division
R.H. Coats Building, 24th Floor, Section E

Ottawa, Ontario, Canada K1A 0T6
(613) 951-1284 Fax (613) 951-5403

Leonard.landry@statcan.ca

Canada

